
  El presente tutorial se orienta a introducir a Holtek a los ingenieros y developers familiarizados con otras tecnologías,
de unaforma prácticay concisa. Primeroanalizaremosla arquitecturade estosmicrosinteresantesy luegoharemos
algunascomparaciones con microsconocidos.Finalmente,haremosuna brevedescripción, conejemplos,del usode
los periféricos más comunes.

  Presentamos a continuación la arquitectura de los micros de las series “Cost Effective”, “I/O” y “A/D”, de Holtek.
La arquitecturadeHoltek esmuy similar a la de PIC, sin embargo, en aquellos lugares enlos quelos usuarios deotros
micros sesientenincómodoscon PIC,Holtek resultaagradable.La siguiente esunadescripciónenumerandoaquellos
factores arquitectónicos standard que facilitan la comprensión:�

Sebasa en laarquitecturaHarvard, dado que dispone deunamemoria de instrucciones,y unade datos, conbuses
separadosy sin comunicaciónentre sí. Consecuenciade estaarquitecturaes que enun llamado asubrutinao
interrupción,el procesadorno puedeguardarla dirección de retorno en la memoria de datos,paralo cual requiere
deun stackindependiente,lo cual limita la cantidadde llamadasanidadasa la cantidadde niveles presentes en el
hardwarestack (2, 4, etc),comomuchos DSP. Por esta mismarazón,estosprocesadoresno poseenintrucciones de
manejo de stack (PUSH, POP, etc).�
Empleapipelining, todas las instruccionesdemoranun ciclo de máquina(cuatrociclos de clock) en ejecutarse,
excepto aquellas que obligan a purgar el pipeline, como instrucciones de salto.�
Utiliza técnicas deVLIW al incorporarlos datos en lainstrucción,es decir, la memoria de instrucciones es deun
ancho mayor, demodode poderincluír constantes(valoresinmediatos,direccionesde datos) en las instrucciones.
Por ejemplo, muchos micros de esta serie tienen un ancho de palabra de instrucciones de 14-bits. En este espacio se
distribuyen las más de 60 instrucciones disponibles,algunas de las cuales incluyenlos 8 bits del dato
correspondiente (direccionamientoinmediato)o su dirección en memoria de datos (direccionamiento directo). Esto
limita la cantidad contigua de memoria de datos disponible a 256 bytes� 28 � .�
La ALU puederealizar operacioneslógicas y aritméticassobrecualquier variable en memoria de datosy el
acumulador (registro A). El resultado de la operación seespecificaen el acumuladoro la memoria de datos, según
la operación;estoatenúael inconveniente de disponer desólo un registro en laCPU (respectode otros RISC y
algunos CISC), dado quepuedehacerse queuna determinadaoperación no altere elacumulador,guardandoel
resultado en la memoria de datos.�
Sóloesposibleescribir en memoria de datosdesdela ALU (el resultado deunaoperación)o el acumulador,no es
posiblecargarun dato inmediatoen memoria sino que es necesariocargarloprimero en el registro Ay luego
cargar éste en memoria.�
Dispone de registrospara funcionesespeciales(SFR, Special FunctionRegisters)mapeados en la memoria de
datos, el registro deestado(STATUS)de laCPU,el acumulador (ACC)y la partebaja delProgramCounter(PCL)
se vencomoun SFRmás. Elestadode la última operación seobtieneaccediendoa la posiciónde STATUS en
memoria de datos, como una variable más.�
La memoria de datos se direcciona solamente de formadirecta, dado quecarecede punteros; sin embargo,
medianteel usodeun par deSFRs,permite elaccesoindirecto a la memoria de datos deunaforma no demasiado
complicada:un SFRcontiene la dirección aaccedery otro "representa"la operación en sí, es decir, el valor del
registro MP(MemoryPointer) es la dirección del dato,y al operarsobreel registroIAR (IndirectAccessRegister)
se operasobreel dato en la dirección queindica FSR. Algunos microstienen dos juegosde estospunteros:
(MP0/IAR0, MP1/IAR1).�
La memoria de instrucciones es de 14-bits, 15-bits,o 16-bits deancho,y hacenfalta algunos deellosparaindicar
que lainstrucciónes de control deflujo de programa(call, jump, etc) ,empleándoseel restoparaespecificarla
dirección. El ancho depalabradeprogramaes coincidente con la capacidad de memoria deprograma,por lo que
la totalidad de la misma es direccionable linealmente.�
Parapermitir mayor versatilidaden el control delflujo del programa,el programcounter estáaccesibleen la
memoria de datos, esun SFRmás. Dado que la memoria de datos es de 8-bits,los bits (14, 15ó 16) del PC están
partidos enPCL (8-bits) y PCH (6,7 ú 8 bits). ElPCH no esaccesibledirectamente,por lo que las instrucciones
que modifiquen elflujo mediantecálculo y operaciónsobrePCL (por oposiciónal uso de CALL o JMP) están
limitadas a un segmento de 256 words de programa.�
Carece de direccionamiento relativo, al cualreemplazapor un "saltearsi" (skip), es decir, laCPU "evita" la
instrucciónsiguienteantedeterminadacondición. Por estarazón,paraejecutaroperacionescomunescomosaltarsi
la última suma diódesborde(carry) o el resultadofue cero,debehacerseun chequeoexplícito del flag en elSFR
correspondiente (STATUS) mediante una instrucción skip.�
Todoslos periféricosse vendesdela CPU comoSFRs.Es posiblemodificar muchascaracterísticasoperativas de
cada uno deellos simplementeescribiendoen posicionesfijas de memoria de datos. Deigual modo, algunas
características de funcionamiento de la CPU (habilitación de interrupciones) se modifican mediante SFRs.�
La arquitecturapermiteinterrupcionesvectorizadas, aunque enrealidadsetratade offsetsfijos. Una interrupción
ocasionaun salto aunaposiciónfija en memoria de instrucciones, dependiente de la causa de lainterrupción.El
contenido del PC seguardaen el hardwarestackpara permitir su recuperación al ejecutaruna instrucciónde
retorno,que restableceel estadode lasinterrupciones.Dado que no existen instrucciones de manejo de stack,
tampocose salva elcontextode laCPU,el programadebesalvar el registro Ay el STATUSenalgunaposiciónde

memoria.Comosólo las operacionesaritméticasy lógicas afectan elSTATUS,esposibleprescindirde salvareste
registro si el interrupt handler realiza funciones clásicas.

� � � � � 	 � � � � � � � � � � � � � � � � � � � � � �

Haremosa continuaciónunacomparación conun CISC bienconocido,comopor ejemploel 68HC908de Motorola,
comunmente conocido como HC08.

Como buen CISC, elHC08 puedeoptimizar el uso de memoria si elprogramadordomina la totalidad de las
instruccionesy susdiversosmodosde direccionamientoparacadauna,generalmentediferentesentresí dado elpoco
grado de ortogonalidad de la CPU y su baja cantidad de registros.

Como es un RISC simple, Holtek requiere más instruccionespara realizar lo mismo, pero estas instrucciones
adicionales nosignificanmásciclos de ejecución,ni siquieramáspalabrasde código,sino todo lo contrario:Holtek
resuelvelo mismo en menosciclos de clock y con menospalabrasde código.Las instrucciones son simples;todo es
cuestión depensarenRISCenvezde CISC, reduciendo el problema aun conjuntodeoperacioneselementales envez
de buscar cuál instrucción optimiza el acceso necesario para resolver el problema.

Motorola HC08 series JL3 y QY4

Presenta direccionamientolineal hasta4K flash y 128 bytesRAM. Modelosmás avanzados de otras seriespermiten
direccionamiento lineal hasta 60K flash y 2K RAM.

Es un CISC dearquitecturaVon Neumann,con stack enRAM, lo que posibilita ilimitados llamados asubrutinase
interrupcionesanidadas.Las interrupcionesson vectorizadas (existeun vectordonde seguardala dirección en la que
seencuentrael handlerparaperiféricoo condición deinterrupción).La CPUposeetres registros: A,H y X, aunque en
realidadH sólopuedeaccedersea través del stacko comopartealtadeHX, un registro índice de 16-bits. El registro X
puedeutilizarsecomoíndice de 8-bits uoffset de 8-bitsrespectode un índice fijo. Todas lasoperacionesaritméticas
sonrespectodel acumulador (A),y el destino de toda operaciónaritméticaes elacumulador.La arquitecturapermite
moverdatosentrevariablesy cargardatos inmediatos en memoria conunasimpleinstrucción(MOV). El clock acepta
un cristal de hasta32 MHz, internamentedividido por cuatro, lo cual dauna ejecucióna 8MIPS pico; pero dada la
cantidaddeciclosdeclock por instrucciónempleados,sepuedeestimarunaoperaciónmáximasostenida de menos de
3 MIPS.

Como periféricos,encontramosun timer de 16-bits condos registros decapturay comparación (CCRs), con
capacidad de generación de PWM, y un conversor AD de 8-bits de ocho canales.

Holtek series “Cost Effective”, “I/O” y “A/D”
Ya hemosanalizadola arquitecturaen el párrafoanterior.Resumiendo,presentadireccionamientolineal hasta8K

OTP y 224 bytes RAM; el modelo que incorpora una cantidad mayor de RAM emplea bank switching.
Setratadeun RISCconarquitecturaHarvard modificada, stack enhardwarede 2 a 16 niveles, según elmodelo.Su

operación es bastantesimilar a PICmicro de Microchip, pero sin eluso de bank switching en flashni I/O y
practicamentenulo en RAM. La CPU aceptaun clock de hasta8 MHz, internamentedividido por cuatro, lo que
permiteunaejecucióna 2MIPS pico, que no disminuyen demasiado en operación sostenida, dado que lagranmayoría
de las operaciones se ejecutan en un ciclo de clock.
Comoperiféricos,encontramosun timer de 8-bits entodoslos modelos;y ninguno,unoo dostimersde 16-bits en las

series “I/O”y “A/D”. Los timerstienencapacidad decontrolarun buzzermediantedossalidascomplementariascon
50% deciclo de trabajo.Los modelosde la serie “A/D”incorporanun conversorAD de 9ó 10-bits de 4 ú 8 canales,y
los modelosmás avanzadosagreganinterfaz I2C y generadorde PWM. El watchdogtimer poseeun oscilador
independiente.

Paralelo de hardware
  La siguiente es una tabla comparativa de características similares de hardware

Motorola Holtek sin AD Holtek con AD

JK1: 1.5K flash, 128 RAM, timer
16-bits, AD, 20 pines, 14 I/O + INT,
xtal, versión RC
QY2: ídem, 16 pines, 14 I/O osc
QT1: ídem, 8 pines, 6 I/O, no AD

48R06/5: 1/0.5K OTP, 64/32 RAM,
timer(8), 16/18 pines, 11/13 I/O,
xtal o RC

46R46/7: 1/2K OTP, 64 RAM, timer(8),
9-bit AD, 18 pines, 13 I/O, xtal o RC

Motorola Holtek sin AD Holtek con AD

JK3: idéntico al JK1, 4K flash
QY4: idéntico al QY2, 4K flash

48R10/30: 1/2K OTP, 64/96 RAM,
timer(8),24/28 pines, 21/25 I/O, osc

46R47: 2K OTP, 64 RAM, timer(8), 9-
bit AD, 18 pines, 13 I/O, xtal o RC

JL3: idéntico al  JK3, 28 pines, 22
I/O + INT

48R50: 4K OTP, 160 RAM, timers
(8,16), 28/48 pines, 15/35 I/O, osc

46R22/3: 2/4K OTP, 64/192 RAM,
timer(8,16), 9/10-bit AD, 24/28 pines,
19/23 I/O, xtal o RC
46R24: 8K OTP, 384 RAM, timers
(16,16), 10-bit AD, 28/48 pines, 20/40
I/O, xtal o RC

Existenademás lasversiones48E quereemplazanla memoria OTP delos 48Rpor memoria MTP (Multiple Times
Programmable) eincorporanmemoria EEPROM y en algunoscasosun segundo punteropara direccionamiento
indirecto.Debidoa que esta tabla no seactualizafrecuentemente, esposibleque existamayor variedad, nodeje de
consultar a su vendedor.

Paralelo de software

A continuaciónharemosun paralelo desoftware.Dijimos que comobuen CISC, elHC08puedeoptimizarel usode
memoria si elprogramadordominala totalidadde las instruccionesy susdiversosmodosde direccionamiento;y que
Holtek, como es un RISC simple, requiere más instrucciones elementalespara realizar lo mismo, peroresuelvela
mismatareaen menosciclosdeclock y con menospalabrasdecódigo.Esto es así porque enun CISCcomoel HC08,
una simple instrucción puede llevar varias operacionescomplejasimplícitas, que requierende varios bytes de
programay varios ciclos de clock para ejecutarse;mientrasque enun RISC como Holtek, las instrucciones son
simplesy el operando siempre está en lainstruccióny seejecutacasi siempre enun ciclo declock.Comodijimos, todo
es cuestión de pensar en RISC, reduciendo el problema a un conjunto de operaciones elementales.

Las comparaciones deusode memoria deprogramaseránuno a uno. Si bienMotorolaempleabytesy Holtek words,
comparamosambosprocesadorespor su capacidad en unidades elementales (1Kbyte vs. 1K word).Lo hacemos de
esta forma dado que la capacidad de memoria deprogramade ambosestáespecificadade esta forma,y nos permite
realizar una comparación directa en precioy prestaciones.Como veremos,generalmenteHoltek emplea menos
unidades elementales para resolver la misma operación.

En cuanto a lavelocidad,a igual valor decristal correspondeigual frecuencia declock e igual cantidadde MIPS
pico, por lo que la comparación enciclos es equivalente. Sin embargo,paraser justos,Motorola admite un clock
cuatrovecesmayor(32 MHz vs 8 Mhz), lo cual puedeapalearla mayorcantidaddeciclosnecesariospararesolverla
misma operación, en la mayoría de los casos.
Los listados acontinuaciónpresentanprimerola resolución deunatareatípica enprogramassimples de control (a lo

que estosmicros están destinados)para el HC08, y luego el equivalente enHoltek. De este modo, el usuario
experimentado enun micro conocidocomo el HC08 puedetener una idea rápida de las capacidades deHoltek.
Respectoal assembler deHoltek, el mismo determinamediantelas declaraciones qué nombres corresponden a
variablesy cuáles setrata de constantes,generandoautomáticamentedireccionamiento directoo inmediatosegún
corresponda. Ante una ambigüedad, existe la directiva OFFSET

Interrupt desde un timer

  Evaluamos el tiempo necesario en atender una interrupción de un timer, salvar contexto, y retomar ejecución normal
del programa.

[03] lda TSC
[04] bclr 7,TSC             ; Ack TOF int
[02] pshh ; salva H

...
[02] pulh ; recupera H
[07] rti

18 ciclos + overhead del procesador

mov somewhere,A ; salva A
mov A,STATUS
mov somewhere2,A ; salva STATUS
...
mov A,somewhere2
mov STATUS,A
mov A,somewhere
reti

8 ciclos + overhead = 10 ciclos

Bifurcación (branch) según flag

Este es el típicocasodemodificacióndel flujo deprogramadebidoa circunstanciasespecificadasenun flag. Dada la
carenciade direccionamiento relativo,Holtek lo resuelvemedianteuna instrucción de salteo (skip)y un salto.
Generalmente, la bifurcación debe pensarse al revés

[05] brclr bit,var,l1
...

l1

5 ciclos, 3 bytes

snz var.bit ; saltea instrucción si el bit 'bit' de la variable 'var' es 1
jmp l1 ; caso contrario salta a 'l1'
...

l1:

2/3 ciclos, 2 words

Salteo (skip) según flag

Esta situación seproducegeneralmentecuandodebemosalterarel estadodeun pin según nosindicaun flag, pero no
podemosdarnosel lujo de producir glitches(modififcarlo primeroasumiendoun valor y despuéscorregirlo si el flag
indica otra cosa). Consideramos sólamente el tiempo empleado en la decisión.

[05] brclr bit,var,l1
; acción si 1

[03] bra l2
l1 ; acción si 0
l2 ; continúa

5/8 ciclos, 5 bytes

sz var.bit
; acción si 1
snz var.bit
; acción si 0

3 ciclos, 2 words

  En un caso más simple, cuando sí nos podemos permitir el lujo de un glitch o simplemente modificamos un flag:
; acción si 0

[05] brclr bit,var,l1
; acción si 1

l1 ; continúa

5 ciclos, 5 bytes

; acción si 0
sz var.bit
; acción si 1

2 ciclos, 1 word

Modificar flag

  Alteramos el estado (seteamos o reseteamos) de un bit en una variable

[04] bset bit,var

4 ciclos, 2 bytes

set var.bit

1 ciclo, 1 word

Típico loop

Consideramos solamente el overhead que introduce lo necesarioparaproducir el loop, es decir, decremento del
contadory salto al inicio delloop. Parael casode HC08,evaluaremosun alternativacon el contador en memoriay
otra con el contador en un registro

l1
...

[05] dbnz var,l1

contador en memoria, 5 ciclos, 3 bytes 

[03] dbnza l1

contador en registro,3 ciclos, 2 bytes 

l1:
...
sdz var ; decrementa 'var' y si ésta es 0, saltea la siguiente instrucción
jmp l1 ; vuelve al loop

3 ciclos, 2 words

Sumar valor a variable 8 bits

Sumamosun determinadovalor constante auna variablecualquiera;el resultadodebequedar en la variable en
cuestión.

[03] lda var
[02] add #value ; resultado de la suma en A
[03] sta var

8 ciclos, 6 bytes

mov A,value
addm A,var ; resultado de la suma en var

2 ciclos, 2 words

Sumar una variable a otra, 8 bits

  Sumamos dos variables entre sí; el resultado debe quedar en la última variable direccionada.

[03] lda var1
[03] add var2 ; resultado de la suma en A
[03] sta var2

9 ciclos, 6 bytes

mov A,var1
addm A,var2 ; resultado de la suma en var2

2 ciclos, 2 words

Memory move, inmediato 8 bits

  Colocamos un valor constante en una posición de memoria.

[04] mov #value,var

4 ciclos, 3 bytes

mov A,value
mov var,A

2 ciclos, 2 words

Memory move, directo 8 bits

  Colocamos en una variable el valor contenido en otra.

[05] mov var1,var2

5 ciclos, 3 bytes

mov A,var1
mov var2,A

2 ciclos, 2 words

Memory move, inmediato 16 bits

  Colocamos un valor constante en una posición de memoria, esta vez en 16-bits.

[03] ldhx #number
[04] sthx var16

7 ciclos, 5 bytes

mov A,LOW number
mov varl,A
mov A,HIGH number
mov varh,A

4 ciclos, 4 words

Memory move, directo 16 bits

  Colocamos en una variable el valor contenido en otra, esta vez en 16-bits.

[04] ldhx var16_1
[04] sthx var16_2

8 ciclos, 4 bytes

mov A,var1l
mov var2l,A
mov A,var1h
mov var2h,A

4 ciclos, 4 words

Cálculo en 16 bits: K = next_T - advance – COMP_K

Realizamosun pequeño cálculono muy complejo,en el cualobtenemosel valor deuna variable deprocesoque
depende de otrasdosvariablesy una constante de calibración. Las variables son todas de 16-bits,y la constante de
calibración es de 8 bits,paramostrarun cálculocombinado.Recordemosque enMotorola los números en 16-bits se
almacenan byte bajo – byte alto.

[03] lda next_T+1
[03] sub advance+1                   ; K = next_T – advance (LSB)
[01] tax   ; a X
[03] lda next_T                      ;
[03] sbc advance   ; MSB

[02] psha   ; a H: HX =   next_T – advance
[02] pulh
[02] aix #COMP_K                     ; ...  - COMP_K (8 bits)
[04] sthx K

23 ciclos, 15 bytes

mov A,next_Tl
sub A,advancel ; next_T – advance (LSB)
mov Kl,A
mov A,next_Th
sbc A,advanceh ; MSB
mov Kh,A
mov A,Kl
sub A,COMP_K ; - COMP_K (8 bits)
mov Kl,A
snz C ; borrow ?
dec Kh ; sí, Kh = Kh – 1

done:

11 ciclos, 11 words

Máquina de estados

  Analizamos a continuación un dispatcher del tipo comunmente utilizado en máquinas de estados, en el cual el flujo
de programa es redirigido según el valor de una variable, es decir, el estado de la máquina.

[01] clrh
[03] ldx STATE ; estados posibles: 0,3,6,9, etc
[06] jsr STTAB,X

STTAB
[03] jmp State1
[03] jmp State2

10 + 3 ciclos, 3 bytes por estado

mov A,STATE ; estados posibles: 0,1,2,3,4,etc
addm A,PCL ; resultado de la suma en PCL
jmp State1
jmp State2

5 ciclos, 1 word por estado

Table read, tabla en RAM

Tomamosun valor deuna tabla según nosindica unavariable queoficia de índice dentro de la tabla. La tabla se
encuentra en RAM.

[01] clrh
[03] ldx OFFSET

[03] lda TABLA,x             ; (1-byte table address)

7 ciclos, 5 bytes, resultado en A

mov A,OFFSET TABLA ; tabla
add A,OFFST ; + offset
mov MP,A ; indirecto
mov A,IAR

4 ciclos, 4 words, resultado en A

Table read, tabla en ROM

Tomamosun valor deuna tabla según nosindica unavariable queoficia de índice dentro de la tabla. La tabla se
encuentraen ROM. En el casode Holtek, la tablapuedeocuparun espaciodentro del segmento de 256bytesen
memoria deprogramaen el que se estáejecutandoel código,o en el último delmapade memoria.Si la tabla no se
halla al principio del segmento, el valor deCOMP compensa esta diferencia; de esta forma mantenemosgeneralidad
en la comparación.

[03] ldx OFFSET ; asumo H=0
[04] lda TABLA,X           ; (2-byte table address)

7 ciclos, 5 bytes, resultado en A

mov A,OFFST
add A,COMP
mov TBLP,A
tabrdl var ; tabla en última página de 256 bytes memoria de programa

5 ciclos, 4 words, resultado en var (si desea en A, var=ACC)

Direccionamiento indirecto en RAM

  Tomamos un valor de un buffer en RAM e incrementamos el puntero

[02] ldx #tabla
[02] lda ,X
[01] incx

5 ciclos, 4 bytes

mov A,tabla
mov MP,A
mov A,IAR
inc MP ; increment memory pointer

4 ciclos, 4 words

Conversión binario a BCD

Finalmente,evaluaremosunasubrutinamuy común,comola empleadaparapresentarresultados enun displayde 7-
segmentoso LCD alfanumérico, es decir, latradicional conversiónde un byte a BCD. Parael casodel HC08, la
llamamos con el valor a convertir en el registro X.Parael casode Holtek, necesitaremosunaposiciónde memoria
adicional para cargar el valor a convertir (SOURCE).

 [04] BINBCD mov #8,COUNTER ; LOOP COUNTER
 [01] clra ; 0 -> RESULT
 [03] sta RESULT
 [01] L1 lslx ; MSB -> carry
 [03] adc RESULT ; x2
 [02] daa ; pero en decimal
 [03] sta RESULT
 [05] dbnz COUNTER,L1 ; listo?
 [04] rts ; Sí, resultado en A y RESULT
124 ciclos, 16 bytes

BINBCD: mov A,8
mov COUNTER,A ; COUNTER=8
clr RESULT ; RESULT=0

L1: rlc SOURCE ; rota SOURCE a la izquierda (MSB a Cy)
mov A,RESULT ; no afecta Cy
adc A,RESULT ; suma A + RESULT + Cy, resultado en A (A=2 x RESULT + Cy)
daa RESULT ; ajusta A para que sea BCD válido, resultado en RESULT
sdz COUNTER ; decrementa 'COUNTER' y si éste es 0, saltea la siguiente instrucción
jmp L1 ; vuelve al loop
ret ; resultado en RESULT

60 ciclos, 10 words

� � 	 � � � 	 � � �  

I/O ports
Al leer un port desalida,Holtek devuelveel estadodel latch, comola mayoría delos microcontroladores(excepto

PIC). El funcionamiento de ambos es similar

Motorola Holtek

Port PORTA PA

Control DDRA PAC (1=entrada)

Pull-up PTAPUE option memory (todos los ports)

Ej.: Uso del port A: bits 2 y 3 entradas, resto salidas

  Configuramos los bits y mostramos cómo esperar un cambio de estado y setear un determinado estado en el port

INIT: clr PA ; prepara estado bajo en salidas
mov A,00001100b
mov PAC,A ; configura entradas y salidas

WLOA2: sz PA.2 ; espera que PA.2 esté en 0
jmp WLOA2

WHIA3: snz PA.3 ; espera que PA.3 esté en 1
jmp WHIA3

set PA.0 ; pone PA.0 en 1

clr PA.0 ; pone PA.0 en 0

8-bit Timer

El timer de 8 bits permitegenerarinterrupcionesal haceroverflow. Poseeun prescaler de 8 etapas, lo cual permite
manejartiemposrazonablespara eventosdel mundo exterior.Puedetomar clock del sistemao externo, el cual es
divisible por el prescaler poruna potencia de 2 (2 a 256);estocual permite su funcionamientocomotimer o como
contador deeventos.Poseeademásun tercermodode funcionamiento en el cualpuedecontar(clock interno)mientras
está activa laentradade clock externo, lo cual permiterealizarmedición de ancho de pulsos. El mismo registro que
lleva las cuentas puede ser inicializado a un determinado valor, para lograr interrupciones a la frecuencia deseada.

TMR: registro de cuentas, puede leerse o cargarse con un valor determinado
TMRC: registro de control, configura el modo de trabajo y prescaler

Poseeademásun parde salidascomplementariascon 50% deciclo de trabajo (divisor por 2conectadoal overflow del
contador),orientadasa conectarun buzzer. Estaopcióncompartelos pines con el port B,y sehabilita medianteel
estado de uno de los pines, pudiendo habilitarse el buzzer con una sola instrucción.
La interrupciónpor desbordesehabilita en el registro de control deinterrupcionesINTC, comotodoslos periféricos.
Dicha interrupcióntieneun offsetpropio, es decir, elflujo deprogramase redirecciona aunaposiciónfija en memoria
de programa.

En los micros con más deun timer, los nombres delos registros llevanun númeroluegode TMR, por logeneralal
timer de 8 bits corresponden TMR0 y TMR0C.

Ej.: config del timer para beep a ~4 KHz con clock de 4 MHz

Utilizaremos eltimer paragenerarun beep,realizaremosel control del mismomediantela smacrosBEEPON y
BEEPOFF. El buzzer se halla conectado directamente a BZ y \BZ

MACRO
BEEPOFF:

clr PB.0 ; pone BZ y \BZ a cero
ENDM

MACRO
BEEPON:

set PB.0 ; habilita BZ y \BZ, produce beep
ENDM

INIT: mov A,10010000b ; timer, prescaler=2 (2MHz; 0,5us)
mov TMRC,A
clr TMR
clr PBC.0 ; PB0 y PB1 como salidas
clr PBC.1
BEEPOFF

  Los pines cambiarán de estado a cada overflow del timer:0,5us ! 256" 128us , la frecuencia es
1

256us # 4 KHz

Ej.: config del timer para interrupciones cada ~15 ms con clock de 4 MHz

  Generamos interrupciones periódicas cada aproximadamente 15 milisegundos

org 8
jmp HANDLER

org someplace
HANDLER:

mov somewhere,A ; salva A
mov A,STATUS
mov somewhere2,A ; salva STATUS
... tarea
mov A,somewhere2
mov STATUS,A ; recupera STATUS
mov A,somewhere ; recupera A
reti

INIT: mov A,22 ; 64us*234=14,976ms => 256-234=22
mov TMR,A
mov A,10010111b ; timer, prescaler=256 (4MHz/256=>64us)
mov TMRC,A
set INTC.2 ; habilita int del timer
set INTC.0 ; habilita interrupciones

Ej.: config del timer para medición de ancho de pulso con clock de 4 MHz

  Medimos la duración del ancho de un pulso positivo (entre flanco ascendente y descendente), recibido en el pin
PC1/TMR. Comenzamos la medición llamando a MEASURE.

INIT: mov A,11001111b ; pw, flanco ascendente prescaler=1 (1us)
mov TMRC,A
set PCC.1 ; config PC1/TMR como entrada

MEASURE:
set TMRC.4 ; TON, comienza medición

; el timer comienza a contar cuando ve
; un flanco ascendente y se detiene
; en un flanco descendente

WAIT: sz TMRC.4 ; espera a que el timer pare
jmp WAIT ; si el bit no es zero, salta a WAIT
; aquí disponemos del ancho del pulso (0,25-64us) en TMR

Ej.: config del timer para demora de ~10 ms con clock de 4 MHz

Utilizamos el timer para introducir una demora de unos 10 milisegundos, sininterrupciones,para lo cual
monitoreamos el flag de overflow.

DEMORA: mov A,100 ; 64us*156=9,984ms
mov TMRC,A
mov A,10010111b ; timer, prescaler=256 (64us)
mov TMRC,A
clr INTC.5

WAIT: snz INTC.5 ; chequea TF
jmp WAIT
; continúa el programa

16-bit Timer

El timer de 16 bits es esencialmenteigual al de 8 bits,exceptoque no tiene prescalery la fuente de reloj es la del
sistema dividida por cuatro(igual que el micro). La lectura del registroTMR debehacersebytealto - bytebajo,dado

que el byte bajo se accede a través de un latch, de modo de acceder a los 16 bits con dos operaciones de 8 bits. De igual
modo, la escritura se realiza byte bajo - byte alto.

En los micros con más deun timer, los nombres delos registros llevanun númeroluegode TMR, por logeneralal
timer de 16 bits corresponden TMR1 y TMR1C.

Conversor A/D

En los micros descriptos, elconversores de 9ó 10 bits.En amboscasos,puedeusarsecomouno de 8 bits demanera
muysimple, dado que la información está"justificadaa la izquierda",es decir, elbytealto tiene siemprelos 8 bits más
significativos,mientrasque el byte bajo tiene 1 ó 2 bits, en laposición de bit más significativos. De esta forma,
leyendoel bytealto (ADRH), puedeutilizarseel conversorcomouno de 8 bits;mientrasqueleyendoambos(ADRL-
ADRH), puedeusarsecomo si el dato fuerauna fracción de 16 bits(debenenmascararselos 6 ó 7 bits menos
significativos). Laelecciónde qué pines estánconectadosal conversor,y cuál deelloses el queproveela señalparala
conversiónen curso, serealizamedianteel registro de control(ADCR), el mismo registro controla a suvezel inicio
de laconversión(bit START). El clocking del AD se controladesdeel registroACSR,que permite dividir por 2, 8ó
32 la frecuencia declock. La frecuenciamáximadeclock parael AD es de 1MHz. El ADpuedeinterrumpircuando
termina una conversión.

Ej.: conversión de señal en AN0 (PB0) con cristal de 4 MHz, poleado

Configuramoslos registros de controlparaclockearcorrectamente al AD,y seteamoslos valorescorrespondientes
paraleer elpin AN0. La conversiónseinicia en CONVERT y finaliza cuando el AD pone aceroel bit EOCB en el
registro de control, momento apartir del cual los registrosADRL y ADRH contienen el valor correspondiente al
resultado de la conversión.

INIT: mov A,00001000b ; AN0
mov ADCR,a 
mov a,00000001b
mov ACSR,a ; clk/8 (500KHz)

CONVERT:
clr ADCR.7
set ADCR.7  ; reset A/D
clr ADCR.7 ; start A/D (nueva conversión)

WAIT: sz ADCR.6 ; EOCB
jmp WAIT
; a partir de aquí disponemos del valor en ADRL, ADRH

Ej.: conversión de señal en AN0 con cristal de 4 MHz, por interrupciones

  En este caso, iniciamos el proceso manualmente (START), y a cada fin de conversión el AD generará una
interrupción que procesará su resultado

org 0Ch
jmp HANDLER

org someplace
HANDLER:

mov somewhere,A ; salva A
mov A,STATUS
mov somewhere2,A ; salva STATUS
.. tarea, dato en ADRL-ADRH
clr ADCR.7
set ADCR.7  ; reset A/D
clr ADCR.7 ; start A/D (nueva conversión)
mov A,somewhere2
mov STATUS,A ; recupera STATUS
mov A,somewhere ; recupera A
reti

INIT: mov A,00001000b ; AN0
mov ADCR,A
mov a,00000001b
mov ACSR,a ; clk/8 (500KHz)
set INTC.3 ; habilita A/D interrupt
set INTC.0 ; habilita interrupciones

START clr ADCR.7
set ADCR.7 ; reset A/D
clr ADCR.7 ; start A/D

; 46R23 usa INTC0 en vez de INTC

; 46R24 también, y el "vector" del AD es 010h

